
Bird eye’s view of a parser

sequence
of tokens Parser tree repre-

sentation

• check that the token sequence correspond to a syntactically
correct program

• if yes: yield tree as intermediate representation for subsequent
phases

• if not: give understandable error message(s)
• we will encounter various kinds of trees

• derivation trees (derivation in a (context-free) grammar)
• parse tree, concrete syntax tree
• abstract syntax trees

• mentioned tree forms hang together
• result of a parser: typically AST

Sample syntax tree

Natural-language parse tree

“Interface” between scanner and parser

• remember: task of scanner = “chopping up” the input char
stream (throw away white space etc) and classify the pieces (1
piece = lexeme)

• classified lexeme = token
• sometimes we use 〈integer, ”42”〉

• integer: “class” or “type” of the token, also called token name
• ”42” : value of the token attribute (or just value). Here, it’s

directly the lexeme (a string or sequence of chars)

• a note on (sloppyness/ease of) terminology: often: the token
name is simply just called the token

• for (context-free) grammars: the token (symbol) corrresponds
there to terminal symbols (or terminals, for short)

Grammars

• in this chapter(s): focus on context-free grammars
• thus here: grammar = CFG
• as in the context of regular expressions/languages: language =
(typically infinite) set of words

• grammar = formalism to unambiguously specify a language
• intended language: all syntactically correct programs of a
given progamming language

Slogan
A CFG describes the syntax of a programming language. a

aand some say, regular expressions describe its microsyntax.

• note: a compiler will reject some syntactically correct
programs, whose violations cannot be captured by CFGs.

Context-free grammar

Definition (CFG)
A context-free grammar G is a 4-tuple G = (ΣT ,ΣN , S ,P):
1. 2 disjoint finite alphabets of terminals ΣT and
2. non-terminals ΣN

3. 1 start-symbol S ∈ ΣN (a non-terminal)
4. productions P = finite subset of ΣN × (ΣN + ΣT )∗

• terminal symbols: corresponds to tokens in parser = basic
building blocks of syntax

• non-terminals: (e.g. “expression”, “while-loop”,
“method-definition” . . . )

• grammar: generating (via “derivations”) languages
• parsing: the inverse problem
⇒ CFG = specification

BNF notation

• popular & common format to write CFGs, i.e., describe
context-free languages

• named after pioneering (seriously) work on Algol 60
• notation to write productions/rules + some extra
meta-symbols for convenience and grouping

Slogan: Backus-Naur form
What regular expressions are for regular languages is BNF for
context-free languages.

“Expressions” in BNF

exp → exp op exp | ( exp ) | number
op → + | − | ∗

(1)

• “→” indicating productions and “ | ” indicating alternatives. 1

• convention: terminals written boldface, non-terminals italic
• also simple math symbols like “+” and “(′′ are meant above as
terminals.

• start symbol here: expr
• remember: terminals like number correspond to tokens, resp.
token classes. The attributes/token values are not relevant
here.

1The grammar can be seen as consisting of 6 productions/rules, 3 for expr
and 3 for op, the | is just for convenience. Side remark: Often also ::= is used
for →.

Different notations

• BNF: notationally not 100% “standardized” across books/tools
• “classic” way (Algol 60):

<exp> : := <exp> <op> <exp>
| ( <exp> )
| NUMBER

<op> : := + | − | ∗

• Extended BNF (EBNF) and yet another style

exp → exp ( ” + ” | ”− ” | ” ∗ ” ) exp
| ”(” exp ”)” | ”number”

(2)

• note: parentheses as terminals vs. as metasymbols

Different ways of writing the same grammar

• directly written as 6 pairs (6 rules, 6 productions) from
ΣN × (ΣN ∪ ΣT )∗, with “→” as nice looking “separator”:

exp → exp op exp
exp → ( exp )
exp → number
op → +
op → −
op → ∗

(3)

• choice of non-terminals: irrelevant (except for human
readability):

E → E O E | (E ) | number
O → + | − | ∗

(4)

• still: we count 6 productions

Grammars as language generators

Deriving a word:
Start from start symbol. Pick a “matching” rule to rewrite the
current word to a new one; repeat until terminal symbols, only.

• non-deterministic process
• rewrite relation for derivations:

• one step rewriting: w1 ⇒ w2
• one step using rule n: w1 ⇒n w2
• many steps: ⇒∗ etc.

language of grammar G

L(G ) = {s | start ⇒∗ s and s ∈ Σ∗T}

Example derivation for (number −number)∗number

exp ⇒ exp op exp
⇒ (exp) op exp
⇒ (exp op exp) op exp
⇒ (number op exp) op exp
⇒ (number − exp) op exp
⇒ (number −number)op exp
⇒ (number −number)∗ exp
⇒ (number −number)∗number

• underline the “place” were a rule is used, i.e., an occurrence of
the non-terminal symbol is being rewritten/expanded

• here: leftmost derivation2

2We’ll come back to that later, it will be important.

Rightmost derivation

exp ⇒ exp op exp
⇒ exp op number
⇒ exp ∗number
⇒ (exp op exp)∗number
⇒ (exp op number)∗number
⇒ (exp−number)∗number
⇒ (number −number)∗number

• other (“mixed”) derivations for the same word possible

Some easy requirements for reasonable grammars

• all symbols (terminals and non-terminals): should occur in a
word derivable from the start symbol

• words containing only non-terminals should be derivable
• an example of a silly grammar G (start-symbol A)

A → B x
B → A y
C → z

• L(G ) = ∅
• those “sanitary conditions”: very minimal “common sense”
requirements

Parse tree

• derivation: if viewed as sequence of steps ⇒ linear “structure”
• order of individual steps: irrelevant
• ⇒ order not needed for subsequent steps
• parse tree: structure for the essence of derivation
• also called concrete syntax tree.3

1 exp

2 exp

number

3 op

+

4 exp

number

• numbers in the tree
• not part of the parse tree, indicate order of derivation, only
• here: leftmost derivation

3there will be abstract syntax trees as well.

Another parse tree (numbers for rightmost derivation)

Abstract syntax tree

• parse tree: contains still unnecessary details
• specifically: parentheses or similar used for grouping
• tree-structure: can express the intended grouping already
• remember: tokens contain also attibute values also (e.g.: full
token for token class number may contain lexeme like ”42”
. . . )

1 exp

2 exp

number

3 op

+

4 exp

number

AST vs. CST

• parse tree
• important conceptual structure, to talk about grammars . . . ,
• most likely not explicitly implemented in a parser

• AST is a concrete datastructure
• important IR of the syntax of the language to be implemented
• written in the meta-language used in the implementation
• therefore: nodes like + and 3 are no longer tokens or lexemes
• concrete data stuctures in the meta-language (C-structs,

instances of Java classes, or what suits best)
• the figure is meant as schematic only
• produced by the parser, used by later phases (often by more

than one)
• note also: we use 3 in the AST, where lexeme was "3"
⇒ at some point the lexeme string (for numbers) is translated to

a number in the meta-language (typically already by the lexer)

Plausible schematic AST (for the other parse tree)

*

-

34 3

42

• this AST: rather “simplified” version of the CST
• an AST closer to the CST (just dropping the parentheses):
under certain circumstances nothing wrong with it either.

Conditionals

Conditionals G1

stmt → if -stmt | other
if -stmt → if ( exp ) stmt

→ if ( exp ) stmt else stmt
exp → 0 | 1

(5)

Parse tree

if ( 0 ) other else other

stmt

if -stmt

if ( exp

0

) stmt

other

else stmt

other

Another grammar for conditionals

Conditionals G2

stmt → if -stmt | other
if -stmt → if ( exp ) stmt else_part

else_part → else stmt | ε
exp → 0 | 1

(6)

ε = empty word

A further parse tree + an AST

stmt

if -stmt

if ( exp

0

) stmt

other

else_part

else stmt

other

COND

0 other other

Note
A missing else part may be represented by null-pointers in
languages like Java.

Ambiguous grammar

Definition (Ambiguous grammar)

A grammar is ambiguous if there exists a word with two different
parse trees.

Remember grammar from equation (1):

exp → exp op exp | ( exp ) | number
op → + | − | ∗

Consider:

number −number ∗number

2 resulting ASTs

∗

−

34 3

42

−

34 ∗

3 42
different parse trees ⇒ different4 ASTs ⇒ different5

meaning

Side remark: different meaning
The issue of “different meaning” may in practice be subtle: is
(x + y)− z the same as x + (y − z)? In principle yes, but what
about MAXINT ?

4At least in most cases.

Precendence & associativity

• one way to make a grammar unambiguous (or less ambiguous)
• For instance:

binary op’s precedence associativity
+, − low left
×, / higher left
↑ highest right

• a ↑ b written in standard math as ab:

5 + 3/5× 2 + 4 ↑ 2 ↑ 3 =

5 + 3/5× 2 + 423 =

(5 + ((3/5× 2)) + (4(2
3))) .

• mostly fine for binary ops, but usually also for unary ones
(postfix or prefix)

Unambiguity without associativity and precedence

• removing ambiguity by reformulating the grammar
• precedence for op’s: precedence cascade

• some bind stronger than others (∗ more than +)
• introduce separate non-terminal for each precedence level

(here: terms and factors)

Expressions, revisited

• associativity
• left-assoc: write the corresponding rules in left-recursive

manner, e.g.:

exp → exp addop term | term

• right-assoc: analogous, but right-recursive
• non-assoc:

exp → term addop term | term

factors and terms

exp → exp addop term | term
addop → + | −
term → termmulop term | factor

mulop → ∗
factor → ( exp ) | number

(7)

34− 3 ∗ 42

34− 3− 42

Real life example

Non-essential ambiguity

left-assoc

stmt-seq → stmt-seq ; stmt | stmt
stmt → S

stmt-seq

stmt

S

; stmt-seq

stmt

S

; stmt-seq

stmt

S

Non-essential ambiguity (2)

right-assoc representation instead

stmt-seq → stmt ; stmt-seq | stmt
stmt → S

stmt-seq

stmt-seq

stmt-seq

stmt

S

; stmt

S

; stmt

S

Possible AST representations

Seq

S S S

Seq

S S S

Dangling else

Nested if’s

if ( 0 ) if ( 1 ) other else other

Remember grammar from equation (5):

stmt → if -stmt | other
if -stmt → if ( exp ) stmt

→ if ( exp ) stmt else stmt
exp → 0 | 1

Should it be like this

stmt

if -stmt

if ( exp

0

) stmt

if -stmt

if ( exp

1

) stmt

other

else stmt

other

. . . or like this

stmt

if -stmt

if ( exp

0

) stmt

if -stmt

if ( exp

1

) stmt

other

else stmt

other

• common convention: connect else to closest “free” (=
dangling) occurrence

Unambiguous grammar

Grammar

stmt → matched_stmt | unmatch_stmt
matched_stmt → if ( exp )matched_stmt else matched_stmt

| other
unmatch_stmt → if ( exp ) stmt

| if ( exp )matched_stmt else unmatch_stmt
exp → 0 | 1

• never have an unmatched statement inside a matched
• complex grammar, seldomly used
• instead: ambiguous one, with extra “rule”: connect each else
to closest free if

• alternative: different syntax, e.g.,
• mandatory else,
• or require endif

CST

stmt

unmatch_stmt

if ( exp

0

) stmt

matched_stmt

if ( exp

1

) elsematched_stmt

other

Adding sugar: extended BNF

• make CFG-notation more “convenient” (but without more
theoretical expressiveness)

• syntactic sugar

EBNF
Main additional notational freedom: use regular expressions on the
rhs of productions. They can contain terminals and non-terminals

• EBNF: officially standardized, but often: all “sugared” BNFs
are called EBNF

• in the standard:
• α∗ written as {α}
• α? written as [α]

• supported (in the standardized form or other) by some parser
tools, but not in all

• remember equation (2)

EBNF examples

A → β{α} for A→ Aα | β
A → {α}β for A→ αA | β

stmt-seq → stmt {; stmt}
stmt-seq → {stmt ;} stmt
if -stmt → if ( exp ) stmt[else stmt]

greek letters: for non-terminals or terminals.

42 / 54



Syntax diagrams

• graphical notation for CFG
• used for Pascal
• important concepts like ambiguity etc: not easily recognizable

• not much in use any longer
• example for unsigned integer (taken from the TikZ manual):

uint . digit E

+

-

uint

The Chomsky hierarchy

• linguist Noam Chomsky [Chomsky, 1956]
• important classification of (formal) languages (sometimes
Chomsky-Schützenberger)

• 4 levels: type 0 languages – type 3 languages
• levels related to machine models that generate/recognize them
• so far: regular languages and CF languages

Overview

rule format languages machines closed
3 A→ aB , A→ a regular NFA, DFA all
2 A→ α1βα2 CF pushdown

automata
∪, ∗, ◦

1 α1Aα2 → α1βα2 context-
sensitive

(linearly re-
stricted au-
tomata)

all

0 α→ β, α 6= ε recursively
enumerable

Turing ma-
chines

all, except
complement

Conventions
• terminals a, b, . . . ∈ ΣN ,
• non-terminals A,B, . . . ∈ ΣT

• general words α, β . . . ∈ (ΣT ∪ ΣN)∗

Phases of a compiler & hierarchy

“Simplified” design?
1 big grammar for the whole compiler? Or at least a CSG for the
front-end, or a CFG combining parsing and scanning?

theoretically possible, but bad idea:

• efficiency
• bad design
• especially combining scanner + parser in one BNF:

• grammar would be needlessly large
• separation of concerns: much clearer/ more efficient design

• for scanner/parsers: regular expressions + (E)BNF: simply the
formalisms of choice!

• front-end needs to do more than checking syntax, CFGs not
expressive enough

• for level-2 and higher: situation gets less clear-cut, plain CSG
not too useful for compilers

BNF-grammar for TINY

program → stmt-seq
stmt-seq → stmt-seq ; stmt | stmt

stmt → if -stmt | repeat-stmt | assign-stmt
| read -stmt | write-stmt

if -stmt → if expr then stmt end
| if expr then stmt else stmt end

repeat-stmt → repeat stmt-seq until expr
assign-stmt → identifier := expr
read -stmt → read identifier
write-stmt → write identifier

expr → simple-expr comparison-op simple-expr
comparison-op → < | =

simple-expr → simple-expr addop term | term
addop → + | −
term → term mulop factor | factor

mulop → ∗ | /
factor → ( expr ) | number | identifier

Syntax tree nodes

typedef enum {StmtK ,ExpK} NodeKind;
typedef enum {IfK ,RepeatK ,AssignK ,ReadK ,WriteK} StmtKind;
typedef enum {OpK ,ConstK ,IdK} ExpKind;

/* ExpType is used for type checking */
typedef enum {Void ,Integer ,Boolean} ExpType;

#define MAXCHILDREN 3

typedef struct treeNode
{ struct treeNode * child[MAXCHILDREN ];

struct treeNode * sibling;
int lineno;
NodeKind nodekind;
union { StmtKind stmt; ExpKind exp;} kind;
union { TokenType op;
int val;
char * name; } attr;
ExpType type; /* for type checking of exps */

Comments on C-representation

• typical use of enum type for that (in C)
• enum’s in C can be very efficient
• treeNode struct (records) is a bit “unstructured”
• newer languages/higher-level than C: better structuring
advisable, especially for languages larger than Tiny.

• in Java-kind of languages: inheritance/subtyping and abstract
classes/interfaces often used for better structuring

49 / 54



Sample Tiny program

read x; { input as integer }
if 0 < x then { don ’t compute if x <= 0 }

fact := 1;
repeat

fact := fact * x;
x := x -1

until x = 0
write fact { output factorial of x }

end

Same Tiny program again

read x ; { i npu t as i n t e g e r }
i f 0 < x then { don ’ t compute i f x <= 0 }

f a c t := 1 ;
r epea t

f a c t := f a c t ∗ x ;
x := x −1

u n t i l x = 0
wr i t e f a c t { output f a c t o r i a l o f x }

end

• keywords / reserved words highlighted by bold-face type setting
• reserved syntax like 0, :=, . . . is not bold-faced
• comments are italicized

Abstract syntax tree for a tiny program

Some questions about the Tiny grammy

later given as assignment

• is the grammar unambiguous?
• How can we change it so that the Tiny allows empty
statements?

• What if we want semicolons in between statements and not
after?

• What is the precedence and associativity of the different
operators?

53 / 54



References I

[Chomsky, 1956] Chomsky, N. (1956).
: Three models for the description of language.
IRE Transactions on Information Theory, 2(113–124).

54 / 54


	Grammars
	Introduction
	Context-free grammars and BNF notation
	Ambiguity
	Syntax diagrams
	Chomsky hierarchy
	Syntax of Tiny


